# Řešení soustav nelineárních rovnic

## Řešení soustav nelineárních rovnic užitím Groebnerovy báze

**PŘÍKLAD: Určete všechna řešení soustavy rovnic**

$$x^2 + 2y^2 - 2 = 0,$$
$$x^2 + xy + y^2 - 2 = 0.$$

```
> restart;
> f:=x^2+2*y^2-2; g:=x^2+x*y+y^2-2;
```

$$f := x^2 + 2y^2 - 2$$
$$g := x^2 + xy + y^2 - 2$$

```
> plots[implicitplot]([f,g],x=-10..10,y=-10..10,numpoints=10000,scaling=constrained,color=[red,blue]);
```




**1. Klasické řešení**

```
> Res:=solve({f,g},{x,y});
```

$$Res := \{y = 0, x = \mathrm{RootOf}(\_Z^2 - 2)\}, \{y = \mathrm{RootOf}(3\,\_Z^2 - 2), x = \mathrm{RootOf}(3\,\_Z^2 - 2)\}$$

```
> Res1:=allvalues(Res[1]); Res2:=allvalues(Res[2]);
```

$$Res1 := \{y = 0, x = \sqrt{2}\}, \{y = 0, x = -\sqrt{2}\}$$
$$Res2 := \{x = \frac{\sqrt{6}}{3}, y = \frac{\sqrt{6}}{3}\}, \{x = -\frac{\sqrt{6}}{3}, y = -\frac{\sqrt{6}}{3}\}$$

```
> P1:=eval([x,y],Res1[1]); P2:=eval([x,y],Res1[2]);
  P3:=eval([x,y],Res2[1]); P4:=eval([x,y],Res2[2]);
```

$$P1 := [\sqrt{2}, 0]$$
$$P2 := [-\sqrt{2}, 0]$$

$$P3 := \left[\frac{\sqrt{6}}{3}, \frac{\sqrt{6}}{3}\right]$$

$$P4 := \left[-\frac{\sqrt{6}}{3}, -\frac{\sqrt{6}}{3}\right]$$

**2. Řešení užitím Groebnerovy báze**

**Určíme Groebnerovu bázi ideálu generovaného polynomy f, g**

```
> GB:=Groebner[Basis]([f,g],plex(x,y));
```

$$GB := [-2\,y + 3\,y^3, -y^2 + x\,y, x^2 + 2\,y^2 - 2]$$

**V bázi se vyskytuje polynom, který obsahuje pouze neznámou y. Určíme jeho nulové body:**

```
> Reseni_y:=solve(GB[1],{y});
```

$$Reseni\_y := \{y = 0\}, \left\{y = \frac{\sqrt{6}}{3}\right\}, \left\{y = -\frac{\sqrt{6}}{3}\right\}$$

**Jednotlivá řešení pro y postupně dosazujeme do původních rovnic. Dostáváme soustavy dvou rovnic o jedné neznámé x. Ty řešíme.**

**1.** $y = 0$

```
> f_x_1:=eval(f,Reseni_y[1]); g_x_1:=eval(g,Reseni_y[1]);
```

$$f\_x\_1 := -2 + x^2$$

$$g\_x\_1 := -2 + x^2$$

```
> solve(f_x_1,{x}); solve(g_x_1,{x});
```

$$\{x = \sqrt{2}\}, \{x = -\sqrt{2}\}$$

$$\{x = \sqrt{2}\}, \{x = -\sqrt{2}\}$$

**2.** $y = \frac{\sqrt{6}}{3}$

```
> f_x_2:=eval(f,Reseni_y[2]); g_x_2:=eval(g,Reseni_y[2]);
```

$$f\_x\_2 := x^2 - \frac{2}{3}$$

$$g\_x\_2 := x^2 + \frac{1}{3}\,x\sqrt{6} - \frac{4}{3}$$

```
> solve(f_x_2,{x}); solve(g_x_2,{x});
```

$$\left\{x = \frac{\sqrt{6}}{3}\right\}, \left\{x = -\frac{\sqrt{6}}{3}\right\}$$

$$\{x = \frac{\sqrt{6}}{3}\}, \{x = -\frac{2\sqrt{6}}{3}\}$$

**3.** $y = -\frac{\sqrt{6}}{3}$

```
> f_x_3:=eval(f,Reseni_y[3]); g_x_3:=eval(g,Reseni_y[3]);
```

$$f\_x\_3 := x^2 - \frac{2}{3}$$

$$g\_x\_3 := x^2 - \frac{1}{3}x\sqrt{6} - \frac{4}{3}$$

```
> solve(f_x_3,{x}); solve(f_x_3,{x});
```

$$\{x = \frac{\sqrt{6}}{3}\}, \{x = -\frac{\sqrt{6}}{3}\}$$

$$\{x = \frac{\sqrt{6}}{3}\}, \{x = -\frac{\sqrt{6}}{3}\}$$

**PŘÍKLAD: Určete všechna řešení soustavy rovnic**

$$x^2 y^2 - y^4 + 3y - 6 = 0,$$
$$x y + y^2 - 2y - 4 = 0.$$

```
> restart;
> f:=x^2*y^2-y^4+3*y-6; g:=x*y+y^2-2*y-4;
```

$$f := x^2 y^2 - y^4 + 3y - 6$$

$$g := x y + y^2 - 2y - 4$$

```
> plots[implicitplot]([f,g],x=-10..10,y=-10..10,numpoints=10000,scaling=constrained,color=[red,blue]);
```




**Určíme Groebnerovu bázi ideálu generovaného polynomy f, g**

```
> GB:=Groebner[Basis]([f,g],plex(x,y));
```

$$GB := [-10 - 19\,y + 4\,y^2 + 4\,y^3,\ -8\,y^2 + 28 + 5\,x - 3\,y]$$

**V bázi se vyskytuje polynom, který obsahuje pouze neznámou y. Určíme jeho nulové body:**

```
> Reseni_y:=solve(GB[1],{y});
```

$$Reseni\_y := \{y = 2\},\ \{y = \frac{-5}{2}\},\ \{y = \frac{-1}{2}\}$$

**Jednotlivá řešení pro y postupně dosazujeme do původních rovnic. Dostáváme soustavy dvou rovnic o jedné neznámé x. Ty řešíme.**

**1.** $y = 2$

```
> f_x_1:=eval(f,Reseni_y[1]); g_x_1:=eval(g,Reseni_y[1]);
```

$$f\_x\_1 := 4\,x^2 - 16$$

$$g\_x\_1 := 2\,x - 4$$

```
> Reseni_x_1:=solve({f_x_1,g_x_1},x);
```

$$Reseni\_x\_1 := \{x = 2\}$$

**Řešení:**

```
> P1:=eval(eval([x,y],Reseni_x_1),Reseni_y[1]);
```

$$P1 := [2, 2]$$

**2.** $y = -\frac{5}{2}$

```
> f_x_2:=eval(f,Reseni_y[2]); g_x_2:=eval(g,Reseni_y[2]);
```

$$f\_x\_2 := \frac{25\,x^2}{4} - \frac{841}{16}$$

$$g\_x\_2 := -\frac{5\,x}{2} + \frac{29}{4}$$

```
> Reseni_x_2:=solve({f_x_2,g_x_2},x);
```

$$Reseni\_x\_2 := \{x = \frac{29}{10}\}$$

**Řešení:**

```
> P2:=eval(eval([x,y],Reseni_x_2),Reseni_y[2]);
```

$$P2 := \left[\frac{29}{10}, \frac{-5}{2}\right]$$

**3.** $y = -\frac{1}{2}$

```
> f_x_3:=eval(f,Reseni_y[3]); g_x_3:=eval(g,Reseni_y[3]);
```

$$f\_x\_3 := \frac{x^2}{4} - \frac{121}{16}$$

$$g\_x\_3 := -\frac{x}{2} - \frac{11}{4}$$

```
> Reseni_x_3:=solve({f_x_3,g_x_3},x);
```

$$Reseni\_x\_3 := \{x = \frac{-11}{2}\}$$

**Řešení:**

```
> P3:=eval(eval([x,y],Reseni_x_3),Reseni_y[3]);
```

$$P3 := \left[\frac{-11}{2}, \frac{-1}{2}\right]$$

**Závěr: Daná soustava nelineárních rovnic má následující tři řešení**

```
> P:={P1,P2,P3};
```

$$P := \{[2, 2], \left[\frac{29}{10}, \frac{-5}{2}\right], \left[\frac{-11}{2}, \frac{-1}{2}\right]\}$$

**Analogie metody Groebnerovy báze s eliminací soustavy lineárních rovnic**

```
> restart;
> r1:=2*x+3*y-z; r2:=x+y-1; r3:=x+z-3;
```

$$r1 := 2\,x + 3\,y - z$$

$$r2 := x + y - 1$$

$$r3 := x + z - 3$$

```
> Aroz:=linalg[genmatrix]({r1,r2,r3},[x,y,z],flag);
```

$$Aroz := \begin{bmatrix} 2 & 3 & -1 & 0 \\ 1 & 1 & 0 & 1 \\ 1 & 0 & 1 & 3 \end{bmatrix}$$

```
> ArozGJ:=linalg[gaussjord](Aroz);
```

$$ArozGJ := \begin{bmatrix} 1 & 0 & 1 & 3 \\ 0 & 1 & -1 & -2 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> AGJ:=linalg[submatrix](ArozGJ,[1,2],[1,2,3]);
  BGJ:=linalg[col](linalg[submatrix](ArozGJ,[1,2],[1,2,3,4]),4);
```

$$AGJ := \begin{bmatrix} 1 & 0 & 1 \\ 0 & 1 & -1 \end{bmatrix}$$

$$BGJ := [3, -2]$$

```
> NoveRovnice:=linalg[geneqns](AGJ,[x,y,z],BGJ);
```

$$NoveRovnice := \{x + z = 3, y - z = -2\}$$

```
> V1:=plots[implicitplot3d]([r1,r2,r3],x=-10..10,y=-10..10,z=-10..
  10,color=[blue,pink,green],style=patchnogrid):
> V2:=plots[implicitplot3d](NoveRovnice,x=-10..10,y=-10..10,z=-10.
  .10,color=red,style=surfacecontour ):
> plots[display](V1); plots[display](V2); plots[display](V1,V2);
```







>